4. Ingamozgás periodikus külső erő hatására

4.1. Fékezetlen ingamozgás periodikus külső erő hatására

Fékezetlen lineáris matematikai inga

Ha az

$$x'' + k^2 x = 0$$

egyenletre valamilyen periodikus külső erő hat, akkor mozgását a következő differenciálegyenlet írja le:

$$x'' + k^2 x = A \cos \nu t,$$

melynek rendszer alakja:

$$\begin{aligned} x' &= y \\ y' &= -k^2 x + A \cos \nu t. \end{aligned} \tag{1}$$

ahol A a külső hatás amplitúdója, ν a külső hatás körfrekvenciája.

Ismert, hogy az inhomogén egyenlet általános megoldását a homogén általános megoldása és az inhomogén egy partikuláris megoldásaként kapjuk [14].

Tudjuk, hogy a homogén egyenlet általános megoldása:

```
Clear[k];
eqn = {x'[t] == y[t], y'[t] == -k^2 x[t]};
Ha = Simplify[DSolve[eqn, {x[t], y[t]}, t]][[1, 1]]
```

$$x[t] \to C[1]\,\mathrm{Cos}[k\,t] + \frac{1}{k}\,C[2]\,\mathrm{Sin}[k\,t]$$

Az inhomogén egyenlet egy partikuláris megoldása $x(0) = 0$ és $y(0) = 0$ kezdeti értékek esetén, ha $k \neq \nu$:

```
Clear[A, ν];
eqn = {x'[t] == y[t],
   y'[t] == -k^2 x[t] + A Cos[ν t], x[0] == 0, y[0] == 0};
IHp = Factor[Simplify[DSolve[eqn, {x[t], y[t]}, t]],
   Trig → True][[1, 1]]
```

$$x[t] \to \left(2\,A\,\mathrm{Sin}\left[\frac{k\,t}{2}-\frac{t\,\nu}{2}\right]\mathrm{Sin}\left[\frac{k\,t}{2}+\frac{t\,\nu}{2}\right]\right)\Big/\left((k-\nu)\,(k+\nu)\right)$$

Az inhomogén egyenlet általános megoldása tehát az előző kettő összege:

$$C[1]\,\mathrm{Cos}[k\,t]+\frac{1}{k}C[2]\,\mathrm{Sin}[k\,t]+\left(2\,A\,\mathrm{Sin}\left[\frac{k\,t}{2}-\frac{t\,\nu}{2}\right]\mathrm{Sin}\left[\frac{k\,t}{2}+\frac{t\,\nu}{2}\right]\right)\Big/\left((k-\nu)\,(k+\nu)\right)$$

Tetszőleges kezdeti értékekből induló megoldások ($k=1,\ A=1,\ \nu=2$):

Az inhomogén egyenlet néhány megoldása különböző kezdeti értékek esetén:



1. ábra

Ismert, hogy létezik ν frekvenciájú megoldás. Ez az alábbi alakú:

```
eqn = {x'[t] == y[t], y'[t] == -k^2 x[t] + A Cos[ν t],
   x[0] == A / (k^2 - ν^2), y[0] == 0};
IHp = Simplify[Simplify[DSolve[eqn, {x[t], y[t]}, t]],
   Trig → True][[1, 1]]
```

```
x[t] → (A Cos[t ν]) / (k^2 - ν^2)
```

Vegyük észre, hogy ennek a megoldásnak az amplitúdója $\frac{A}{k^2-\nu^2}$, ami arányos a külső erő amplitúdójával és fordítottan arányos $(k^2-\nu^2)$-vel. A hintázás szempontjából tekintve, nagyobb erő nagyobb kilengést jelent, a hinta sajátfrekvenciáját megközelítő külső körfrekvencia is hasonló eredményt ad.

Ugyanakkor ha $\nu \approx k$, a rezgés (lengés) veszélyesen naggyá válik, a lebegés, illetve a rezonancia jelenségével találkozhatunk.

Az inhomogén egyenlet partikuláris megoldása mutatja, hogy egy gyors és egy lassú rezgés összegződése során alakul ki rezonancia, illetve lebegés

```
Simplify[
  DSolve[{x′[t] == y[t], y'[t] == -k^2 x[t] + A Cos[ν t],
    x[0] == 0, y[0] == 0}, {x[t], y[t]}, t]][[1, 1]]
```

```
x[t] → (A (-Cos[k t] + Cos[t ν])) / (k^2 - ν^2)
```

a $\cos(k\, t)$ alakú sajátregzés, illetve a $\cos(\nu\, t)$ alakú gerjesztett rezgés lassan kerül azonos fázisból ellenfázisba. A következő néhány ábra a lebegés jelenségét szemlélteti azokban az esetekben, ha a ν konstanst változtatjuk $k = 1$ körül. Az ábra jól mutatja a rezgések összegződését. A felső ábrákon látható rezgések összegeként létrejövő rezgéseket mutatják az alsó ábrák különböző körfrekvenciájú külső erők esetében ($k = 1$ körüli frekvenciák).

Rezgések összegzése különböző körfrekvenciájú külső erők esetében :



Ha $k = \nu$, azaz a gerjesztés körfrekvenciája azonos a rendszer sajátfrekvenciájával, akkor a megoldások nem korlátosak. Ezt a jelenséget nevezzük rezonanciának. Az egyensúlyi helyzetből induló megoldás a következő alakú:

```
Clear[A, k];
eqn2 = {x'[t] == y[t],
   y'[t] == -k^2 x[t] + A Cos[k t], x[0] == 0, y[0] == 0};
Simplify[DSolve[eqn2, {x[t], y[t]}, t]][[1, 1]]
```

```
x[t] → (A t Sin[k t]) / (2 k)
```

Ezt mutatja a következő ábra:



3. ábra

A rezonancia komoly katasztrófát okozhat. A legismertebb szerencsétlenség az Amerikai Egyesült Államokban következett be 1940. november 7-én.

Az újonnan épített Tachoma-híd egy több napig tartó erős szélben leszakadt. A szél által a hídra gyakorolt periodikus lökések körfrekvenciája megegyezett a híd lengéseinek sajátfrekvenciájával. A kialakult rezonancia miatt a több méter amplitúdójú, és több órán át tartó rezgések a híd leszakadásához vezettek [3].

Interaktív illusztráció

A következő interaktív vizsgálat az inhomogén egyenlet tetszőleges partikuláris megoldásának a külső hatás amplitúdójától való függését szemlélteti.

□ *Javasolt beállítások:*

Ha a kezdeti értékeket 0-ra, a külső erő körfrekvenciáját és a rugalmassági együtthatót 1-re, az amplitúdót tetszőlegesre állítva a kitérés a végtelenbe tart (rezonancia).

Ha most a külső erő körfrekvenciáját 1-hez közeli értékre állítjuk, akkor a lebegés jelensége figyelhető meg.

Ha csupán a külső hatás amplitúdóját változtatjuk interaktívan láthatjuk, hogy a megoldások viselkedését nem befolyásolja az amplitúdóváltozás. A függvénygrafikonok arányosan nőnek, ahogyan az amplitúdó nő.



Interaktív illusztráció

Ha a különböző frekvenciákat az emberi fül által hallható frekvenciatartományból választjuk, akkor az inhomogén egyenlet partikuláris megoldása hallhatóvá válik. Az alábbi animáció két rezgés összegeként létrejövő rezgéseket szemlélteti. Minél nagyobbra állítjuk a frekvenciaértékeket, annál magasabb hangot hallhatunk.

Figyeljük meg a lebegés jelenségét. A hang ereje fokozatosan erősödik illetve elhalkul. Minél közelebbi értékeket választunk, annál jobban megfigyelhető a jelenség a kibocsájtott hang segítségével.

VIGYÁZAT! Ha a két frekvenciaértéket egyformának választjuk, az inhomogén egyenlet partikuláris megoldása a végtelenbe tart, melyet az illusztráció nem képes megjeleníteni.

Javasolt beállítások:

$A = 1$, $k = 1350$, ν értékét pedig változtatjuk k körül.



Fékezetlen nemlineáris matematikai inga

A nemlineáris matematikai inga esetében az egyenlet a következő alakú:

$$\begin{aligned} x' &= y \\ y' &= -k^2 \sin x + A \cos \nu\, t. \end{aligned} \tag{2}$$

ahol A a külső erő amplitúdója, ν a külső erő körfrekvenciája.

A Melléklet 6.4 tétele alapján tudjuk, hogy kis kezdő értékek esetén, ha a homogén, lineáris, periodikus $x'' = -k^2 x$ egyenletnek nincs nemtriviális, ν körfrekvenciájú periodikus megoldása, akkor az (4.2)-nek létezik pontosan egy. Ez az origó környezetében akkor teljesül, ha $\nu \neq k$. Mivel a homogén egyenlet megoldásainak rezgése amplitúdó növelkedésével lassul, ezért nem túl nagy gerjesző amplitúdó (az átfordulást elkerülendő) és adott $\nu > k$ körfrekvencia mellett létezik ν körfrekvenciájú periodikus megoldás.

Interaktív illusztráció

A következő interaktív vizsgálat az inhomogén egyenlet tetszőleges partikuláris megoldásának különböző paraméterektől való függését szemlélteti.

Javasolt beállítások:

Érdemes megfigyelni azt a speciális esetet, amikor minden értéket 1-nek választunk, és a kezdeti kitérést pedig interaktívan változtatjuk, illetve ha a külső hatás körfrekvenciáját változtatjuk, ahol $x(0) = 0$, minden más érték pedig 1. Az első esetben a kezdeti kitérés megváltoztatásával a mozgás jelentős mértékben változik, az utóbbiban pedig megközelítőleg az 1.34-es értékig a kitérés nagy, utánna pedig adott korlátok között mozog.

Ha a külső hatás amplitúdóját változtatjuk, látható, hogy a matematikai ingától eltérően a megoldások viselkedése nagy mértékben változik.



A következő ábrán a javasolt beállítások szerint minden értéket 1-nek választunk, és a kezdeti kitérést változtatjuk.



4. ábra

Szintén a javasolt beállítások szerint a kezdeti kitérést 0-nak választva, minden más értéket pedig 1-nek, a megoldásoknak a külsö hatás körfrekveniájától való függését a következő ábra mutatja:



5. ábra

Az inhomogén egyenlet megoldásának változása különböző amplitúdójú külső erők esetén:



4.2. Fékezett ingamozgás periodikus külső erő hatására

Fékezett matematikai inga

Ha a csillapítást is hozzávesszük (4.1)-hez akkor kapjuk a következő egyenletet

$$\begin{aligned} x' &= y \\ y' &= -a\,y - k^2\,x + A\cos\nu\,t, \end{aligned} \tag{3}$$

ahol A a külső erő amplitúdója, ν a külső erő körfrekvenciája.

Határozzuk meg az inhomogén egyenlet általános megoldását!

A 3. fejezetből tudjuk, hogy a homogén egyenlet általános megoldása, ha feltesszük, hogy a csillapítás kicsi, és alkalmazva a $\mu = \sqrt{a^2 - 4\,k^2}$ helyettesítést:

```
Clear[a, k];
eqn = {x'[t] == y[t],
   y'[t] == -a y[t] - k^2 x[t], x[0] == x0, y[0] == y0};
Ha = Simplify[DSolve[eqn, {x[t], y[t]}, t][[1, 1]] /.
     {√(a^2 - 4 k^2) → ⅈ μ, 1/√(a^2 - 4 k^2) → -ⅈ μ} /.
   ⅇ^u_ :→ ComplexExpand[ⅇ^u]]
```

$$x[t] \to e^{-\frac{a\,t}{2}}\,\mu\left(x0\,\mu\,\mathrm{Cos}\left[\frac{t\,\mu}{2}\right] + (a\,x0 + 2\,y0)\,\mathrm{Sin}\left[\frac{t\,\mu}{2}\right]\right)$$

Láthatjuk, hogy a megoldások nem periodikusak, ezért a Melléklet 6.2 tétele szerint inhomogén egyenletnek lesz periodikus megoldása periodikus külső hatás esetén.

Az inhomogén egyenlet egy partikuláris megoldása $x(0) = 0$ és $y(0) = 0$ kezdeti értékek esetén:

```
eqn = {x'[t] == y[t],
   y'[t] == -a y[t] - k^2 x[t] + A Cos[ν t], x[0] == 0, y[0] == 0};
```

```
IHp = Collect[Simplify[ComplexExpand[
    DSolve[eqn, {x[t], y[t]}, t][[1, 1]] /.
      {√(a² - 4 k²) → ⅈ μ, 1/√(a² - 4 k²) → -ⅈ μ}],
   (a ∈ Reals) && (k ∈ Reals) && (μ ∈ Reals)], ⅇ^(-a t/2)]
```

$$x[t] \to -\left(A\, e^{-\frac{a t}{2}}\, \mu \left(k^2 \mu \operatorname{Cos}\left[\frac{t \mu}{2}\right] - \mu \nu^2 \operatorname{Cos}\left[\frac{t \mu}{2}\right] + a k^2 \operatorname{Sin}\left[\frac{t \mu}{2}\right] + a \nu^2 \operatorname{Sin}\left[\frac{t \mu}{2}\right]\right)\right) \Big/ \left(k^4 + a^2 \nu^2 - 2 k^2 \nu^2 + \nu^4\right) - \left(A \mu \left(-k^2 \mu \operatorname{Cos}[t \nu] + \mu \nu^2 \operatorname{Cos}[t \nu] - a \mu \nu \operatorname{Sin}[t \nu]\right)\right) \Big/ \left(k^4 + a^2 \nu^2 - 2 k^2 \nu^2 + \nu^4\right)$$

Itt a második tag ν körfrekvenciájú, míg az első a homogén egyenlet egy megoldása. A rezonancia jelensége itt is megfigyelhető, ha a gerjesztés körfrekvenciája azonos a rendszer sajátkörfrekvenciájával, azaz $\mu / 2 = \nu$.

Az inhomogén egyenlet megoldásai adott kezdeti értékekre
$(k = 1,\ A = 1,\ \nu = 2)$:



7. ábra

Az inhomogén egyenlet általános megoldása itt is a homogén egyenlet általános és az inhomogén egyenlet egy partikuláris megoldásának összege.

Az inhomogén egyenlet partikuláris megoldása függ a külső hatás amplitúdójától a csillapítatlan esethez hasonló módon ($a = 0.2$, $k = 1$, $x(0) = y(0) = 0$).



8. ábra

A hintázással kapcsolatosan megállapíthatjuk hogy mivel mindig fennáll csillapítás, bármilyen periodikus kis amlitúdójú hintáztatás esetén a hinta periodikusan mozog. Rezonancia lép fel, ha $\mu/2 = \nu$, de ez nem veszélyes (kis amplitúdó mellett, mivel a lengés korlátos marad), csupán a leghatékonyabb hintázást biztosítja.

Interaktív illusztráció

A következő interaktív vizsgálat az inhomogén egyenlet tetszőleges partikuláris megoldásának különböző paraméterektől való függését szemlélteti.

◻ Javasolt beállítások $\mu / 2 = \nu$ rezonancia körül:

$a = 0.1$, $\nu = 0.9987$, $k^2 = 1$, a többi értéket tetszés szerint változtatjuk.


x
10
5
−5
−10
50
100
150
200
t
y
10
5
−5
−10
−10
−5
5
10
x
Csillapítás 0.5
Külső hatás körfrekvenciája 0.9987
Külső hatás amplitúdója 3
Rugalmassági együttható (k2) 1
Kezdeti kitérés 0
Kezdeti sebesség 1
Kicsinyítés 10.
Idő 198.5

Fékezett nemlineáris matematikai inga

A nemlineáris, csillapított nemlineáris matematikai inga egyenletét egészítsük ki egy a rendszerre ható, külső periodikus erővel:

$$\begin{aligned} x' &= y \\ y' &= -a\,y - k^2 \sin x + A\cos \nu\, t. \end{aligned} \tag{4}$$

Kis kezdeti értékek esetén a mozgás a megfelelő matematikai ingáénak a perturbációja és a Melléklet 6.4 tétele szerint létezik ν körfrekvenciájú periodikus mozgás. Ennek következménye a hintázásra hasonló ahhoz, amit a matematikai inga esetén mondtunk. A $\mu/2 = \nu$ esetben rezonancia, a $\mu/2 \approx \nu$ körüli értékek esetén pedig a lebegés figyelhető meg.

Túl nagy kezdeti értékek esetén az inga mozgása kaotikussá válik [11] és [15]. Ezt a jelenséget Hatvani László vizsgálta az alábbi rendszeren:

$$\begin{aligned} x' &= y \\ y' &= -0.1\,y - \sin(x) + \cos(t)\,. \end{aligned}$$

A külső hatás túl nagy amplitúdója, szintén ellenőrizhetetlen mozgást (átfordulásokat) eredményezhet.

Interaktív illusztráció

A lineáris és nemlineáris inga mozgásának összehasonlítása érdekében nézzük a következő animációt, mely mind a csillapított, mind a csillapítatlan esetet szemlélteti. Az animáció segítségével a fenti példát is megvizsgálhatjuk.

A kezdeti kitérést zérusnak tekintve, a kezdeti sebességet pedig 2 kis környezetében változtatva a megoldás aszimpotikusan periodikus 2π-vel (a külső erő periódusával). A kezdeti értékek változtatásával megszámlálhatatlanul sok megjósolhatatlan viselkedésű, kaotikus megoldás létezik.

Az animáció segítségével azt is láthajuk, hogy a kitérés nemcsak a kezdeti értékektől, hanem a külső hatás körfrekvenciájától és amplitúdójától is függ. A nemlineáris matematikai inga kaotikus mozgását a barna színű grafikonok szemléltetik megfelelő paraméterek esetén.

Javasolt beállítások:

$a = 0.1$, $k^2 = 1$, $\nu = 0.9987$. Ebben az esetben $\mu/2 = \nu$. Figyeljük meg mi lesz, ha a ν értéket változtatjuk $\mu/2$ körül!

$a = 0.1$, $k^2 = 1$, $A = 1$, $\nu = 1$, $x(0) = 0$, és $y(0)$-ot változtassuk 2 kis környezetében. Ekkor Hatvani László katotikus ingájának vizualizációját kapjuk.


Lineáris inga
Nemlineáris inga
x
5
0
−5
10
20
30
40
50
60
t
y
5
−5
5
−5
x
Csillapítás 0.1
Rugalmassági együttható (k2) 1
Külső hatás amplitúdója 1
Külső hatás körfrekvenciája 1
Kezdeti kitérés 0
Kezdeti sebesség 2
Kicsinyítés 9
Idő
29.4